
HP3324A Product Note:
 
          A multi-channel setup with programmable inter-channel phase
          ===========================================================
 
Thorough device test can rarely be done insitu. Consideration of repeatability,
limit testing, and safety -quite apart from shear feasibility- are some of the
reasons for simulating input signals rather than trying to force the real
environment to provide all desired situations at will.
 
Automotive, energy and navigation are example segments of the electronic
control device market where these arguments are particularly important.
 
Typical measurement concerns in these segments are: the relative position or
speeds of shafts, power line phase, and the computation of position from signal
phase information. In the past, the problem has been to generate a number of
synchronous signals with programmable phase difference. This product note
explains how this can be solved by master/slaving a number of HP3324A
synthesized function generators.
 
General description
-------------------
 
One HP3324A is equipped with Option 004 (master auto-cal) and the rest with
Option 003 (slave auto-cal). This allows the phase between channels to be set
up accurately on automatic or manual command without external phase measure-
ment. A VHF switch (HP59307A) is required for automatic operation if more than
two channels  are needed. A second switch is necessary for more than five
channels.
 
Features and benefits
---------------------
  -No oscilloscope, phase meter or counter needed for calibration.
     Auto-cal accuracy is +/-1 to +/-7 degree, depending on
     waveform and amplitude mix. Refer to HP3324A data sheet.
  -Manual or programmed operation.
  -Extendable to virtually any number of channels.
  -Output waveforms need not be the same, and are selectable from:
     sine (up to 21 MHz), square or triangle (up to 11 MHz),
     and  ramp (up to 11 kHz). Minimum frequency is 1 mHz.
  -Frequency resolution 9 digits. Stability 5ppm, can be improved
     to 0.05 ppm by fitting master 3324A with Option 001.
  -Independent amplitude 1 mVpp to 10 Vpp into 50 ohm. Option 002
     provides 4 mVpp to 40 Vpp into 500 ohm.
 
Application example: control circuit for energy supply network.
-------------------
To test the design of a control circuit, three signals are needed to represent
the voltages of a 3-phase supply. Additionally, a signal representing current
lead or lag tests the response to power factor changes. As only one current
phase need be simulated at a time, four 3324A’s are adequate in this
application.
 
Setting up for the first time
-----------------------------
After racking or stacking the synthesizers, "daisy chain" the rear panel Refer-
ence connectors, and interconnect the Phase Cal connectors as shown in the
following diagram:

           Master 3324A with Option 004
        ______________  (and Option 001 if high stability needed)
       |              |
       |  Phase Cal In<---------------------------------------
   ---->Oven out (Opt)|                                       |
  |    | Phase Cal Out>----------------------------           |
   ---->Ext.ref.in    |                            |          |
     --<Ext.ref.out   |                            |          |
    |  |______________|                            |          |
    |                                              |          |
    |                                              |          |
    |  1st Slave 3324A with Option 003             |Com A     |Com B
    |   ______________                            _|__________|_
    |  |              |                          |              |
    |  |  Phase Cal In<--< from A1            <--|A1          B1|--<
    |  | Phase Cal Out>--> to B1              <--|A2          B2|--<
     -->Ext.ref.in    |                       <--|A3          B3|--<
   ----<Ext.ref.out   |                       <--|A4          B4|--<
  |    |______________|                          |______________|
  |                                              59307A VHF switch
  |
  |    2nd Slave 3324A with Option 003
  |     ______________
  |    |              |
  |    |  Phase Cal In<--< from A2
  |    | Phase Cal Out>--> to B2
   ---->Ext.ref.in    |
     --<Ext.ref.out   |
    |  |              |
    |  |______________|
    |
    to 3rd slave, etc
 
Now use the following procedure to equalize the path lengths of the master
Phase Cal Output to the various slave Phase Cal Inputs:
 1. Set the following parameters of all 3324As to zero:
       Phase,
       Assign Phase 0,
       Calibration Delay,  }  Select Util menu and roll.
       Signal Delay.       }
 2. Set same frequency, waveform and amplitude on all 3324As.
 3. Press 1st slave’s Shift and Phase Cal keys.
 4. When auto phase cal complete, measure the time interval between correspond-
    ing points on the master and 1st slave outputs. Measure from the master
    signals FORWARDS to the slave signal, even if this means nearly a full
    cycle (this simplifies handling the other slaves).
 5. Enter this value in ps in the 1st slave’s Calibration Delay (Util menu).
 6. Do auto phase cal again, verify phase is OK.
 7. Repeat steps 3 to 5 for the other slaves in turn.
 
Using the setup
---------------
Try to use the same path lengths from the 3324A outputs to the device under
test. If this is not feasible, use a similar procedure to that above, measure-
ing the signals at the DUT, and entering the delay into the appropriate slave’s
Signal Delay (Util menu).
 
From here on, correct phase relationship is simply a matter of setting each
slave to the required phase. Subsequently, whenever frequency, waveform or
amplitude are changed an auto-cal should be performed on each slave.
 
Start-up tools
--------------
An UNSUPPORTED disk is available in order to assist implementation in an
automated environment with the following software:

 1. HP ITG (Interactive Test Generator) applications driver and instrument
    driver IDHP3324PS. These run only on HPBasic platforms, and are NOT
    supported.
 
 2. An HPBasic routine for users to build into their own programs. A listing
    appears at the end of this document. This routine is NOT supported.
 
HP3324A Auto-cal ITG applications driver
----------------------------------------
To load the auto-cal driver, first load the file IDHP3324PS as many times as
you have HP3324As. Each time you will be asked "stand alone or slave" and you
should respond with "slave". Subsequently, change the driver name to "Master",
"Slave1", "Slave2" etc. Now load file ADPC3324A, and enter switch address and
driver names in the first panel as required.
 
The file ADPC3324A generates a small panel (see inset below) that allows the
phase of the master (M) and any slave (Sn) 3324A to be changed, as well as the
common frequency (F). Auto-cal is implemented with the CAL label (an
exclamation mark indicates when this should be done)
     ____________________________________________________________________
    |                      HP Interactive Test Generator                 |
    |File  Instruments...  Applications                              Help|
    |____________________________________________________________________|
    |            |                 MASTER3324,718               |        |
    |            |[Reset] Signal[On] Output[Front]  [Main Panel]|        |
    |            |______________________________________________|        |
    |            |Waveform   [   Sine    ]    Phase [ 0.0 DEG  ]|        |
    |            |                                              |        |
    |            |Offset     [  0   ][ V ]          [Assign P 0]|        |
    |            |                                              |        |
    |            |Amplitude  [  1m  ][Vpp]                      |        |
    |            |                                              |        |
    |            |           [Ampl. Cal. ]  Sig del [   0 ps   ]|        |
    |            |                                              |        |
    |            |Frequency  [  2k       ]  Ref del [ 3000 ps  ]|        |
    |__________________________                                 |        |
    |     PC3324A,0            |                                |        |
    |[Reset] Delay [3000] [Cal]|                                |        |
    |__________________________|                                |        |
    |F[  2 k  ]  M[  0.0 DEG  ]|________________________________|        |
    |                          |    SLAVE3324,718               |        |
    |S1 [32.5DEG]  S2 [   ?   ]|[On] Output[Front]  [Main Panel]|        |
    |                          |________________________________|        |
    |S3 [   ?   ]  S4 [   ?   ]| Sine    ]    Phase [ 32.5 DEG ]|        |
    |__________________________|                                |        |
    |            |Offset     [  0   ][ V ]          [Assign P 0]|        |
    |            |                                              |        |
    |            |Amplitude  [  1m  ][Vpp]                      |        |
    |            |                                              |        |
    |            |           [Ampl. Cal. ]  Sig del [   0 ps   ]|        |
    |            |                                              |        |
    |            |Frequency  [  2k       ]  Ref del [   0 ps   ]|        |
    |            |                                              |        |
    |            |                                              |        |
    |____________________________________________________________________|
 
 
HPBasic Auto-cal routine
------------------------
The routine PHASECAL implements the HP3324A’s built-in phase-cal feature for
setups with up to five function generators connected as shown in the above
diagram.
 
The routine simply switches each slave in turn to the master and triggers the

3324A-internal auto phase cal measurement.
 
1     INTEGER A,B,C,D,E,F  ! this main program is only a dummy program
2     A=0                  ! to be replaced by other main programs
3     B=0
4     C=0
5     D=0
6     E=0
7     F=0
8     INPUT "SWITCH ADDRESS",A,"MASTER ADDRESS",B,"SLAVE1 ADDRESS",C,"SLAVE2 AD
RESS",D,"SLAVE3 ADDRESS",E,"SLAVE4 ADDRESS",F
9     CALL Save_ib_address(A,B,C,D,E,F)
10    CALL Phase_cal
11    END
12    !============================================================
13    SUB Save_ib_address(INTEGER Switch,Master,Slave1,OPTIONAL INTEGER Slave2,
Slave3,Slave4)
14      COM /Hpibdata/ INTEGER Switch_addr,Master_addr,Slave_addr(1:4)
15      INTEGER I
16      MAT Slave_addr= (0)       !  clear adresses
17      Switch_addr=Switch
18      Master_addr=Master
19      Slave_addr(1)=Slave1
20      IF NPAR>3 THEN Slave_addr(2)=Slave2
21      IF NPAR>4 THEN Slave_addr(3)=Slave3
22      IF NPAR>5 THEN Slave_addr(4)=Slave4
23    SUBEND
24    !============================================================
25    SUB Phase_cal
26      COM /Hpibdata/ INTEGER Switch_addr,Master_addr,Slave_addr(1:4)
27      INTEGER I,J
28      REAL Master_phase,Frequency
29      DIM Lf$[2]
30      Lf$=CHR$(13)&CHR$(10)             ! separator for switch commands
31      ON TIMEOUT 7,30 GOTO End          ! device select code 7 for HPIB card
                                            assumed !!!
32      IF Master_addr THEN
33        OUTPUT Master_addr;"IPH"        ! read master phase
34        ENTER Master_addr USING "K";Master_phase
35        OUTPUT Master_addr;"PH 0DEIFR"  ! set phase during calibration to 0
36        ENTER Master_addr USING "K";Frequency   ! read frequency from master
as reference for slaves
37        FOR I=1 TO 4
38          IF Slave_addr(I)<>0 THEN
39            IF Switch_addr THEN         ! connection of slave by switch
40              OUTPUT Switch_addr;"A"&VAL$(I)&Lf$&"B"&VAL$(I)  !  change this
line if using other VHF switch than HP59307A ( connects each slave to master )
41            ELSE                        ! manual connection of selected slave
42              DISP "Please connect SLAVE"&VAL$(I)&" to MASTER instrument, con
firm with any key"
43              ON KBD ALL GOTO Confirmed
44              GOTO 44                   ! wait for keypress
45 Confirmed:   OFF KBD
46              DISP "
           "
47            END IF
48            OUTPUT Slave_addr(I);"FR"&VAL$(Frequency)&"HZ"  ! set slave to sa
me frequency as master
49            OUTPUT Slave_addr(I);"PC"   ! start calibration
50            J=SPOLL(Slave_addr(I))
51            !   if calibration is failed, ( J AND 8 ) is <> 0, a failure rout
ine could be done here ( to ensure failure , use ISE command of 3324A )
52          END IF
53        NEXT I
54        OUTPUT Master_addr;"PH"&VAL$(Master_phase)&"DE"   !  restore phase of

master
55        BEEP 160,.05      ! finished
56      END IF
57 End:    !
58      OFF TIMEOUT
59    SUBEND
 

Page 5


